

HIV and Antiviral Therapy

Robert M. Corless
Ontario Research Centre for Computer Algebra
<http://www.orcca.on.ca>

An exploration of "Mathematical Models and the design of public health policy: HIV and Antiviral therapy", by Sunetra Gupta, Roy M. Anderson, and Robert M. May, SIAM Review, Vol 35, No. 3, pp. 1-16, March 1993.

Notation of the paper:

p = proportion of cohort

$\mu * N[0]$ = sexually active population

$V[x]$ = vaccinated susceptibles

X = unvaccinated susceptibles

s = per capita rate of vaccination

λ = per partnership probability of loss

c = mean rate of partner change

$N[0]$ = stable population size in the absence of infection

μ = inverse of average duration of sexual activity

Y = untreated infected population

v = per capita rate of development of AIDS

$V[y]$ = number of infected individuals receiving therapy

d = per capita rate of treated infected population development of AIDS

$B[1]$ = probability of transmission from an untreated carrier of HIV

$B[2]$ = probability of transmission from a treated carrier.

N = current size of the sexually active population

> **restart;**

The assumptions of the exposition in the paper (we can come back and change this later and see how the conclusions change):

> **s := 0;**

$$s := 0$$

> **r := 0;**

$$r := 0$$

The basic equations in differential form (with the dependence on time suppressed for ease of working with equilibria): the rate of change of unvaccinated susceptibles depends on how many come of age, and how many are infected, and how many stop sexual activity, and how many are vaccinated:

> **eq1 := diff(X(t),t) = mu*N[0]*(1-p) - (c*lambda + mu + s)*X;**

$$eq1 := \frac{\partial}{\partial t} X(t) = \mu N_0 (1-p) - (c \lambda + \mu) X$$

Similarly, the rate of change of vaccinated susceptibles depends on how many come of age and are vaccinated, and how many are infected, and how many stop sexual activity, and how many are vaccinated from the X population.

> **eq2 := diff(V[x](t),t) = mu*N[0]*p-(c*lambda+mu)*V[x] + s*X;**

$$eq2 := \frac{\partial}{\partial t} V_x(t) = \mu N_0 p - (c \lambda + \mu) V_x$$

> **eq3 := diff(Y(t),t) = c*lambda*X-(v+r+mu)*Y;**

$$eq3 := \frac{\partial}{\partial t} Y(t) = c \lambda X - (v + \mu) Y$$

> **eq4 := diff(V[y](t),t) = c*lambda*V[x] + r*Y - (d+mu)*V[y];**

$$eq4 := \frac{\partial}{\partial t} V_y(t) = c \lambda V_x - (d + \mu) V_y$$

The paper doesn't say so, but the calculation only works out if we say that the total population is just the sum of the parts.

> **NN := X + V[x] + Y + V[y];**

$$NN := X + V_x + Y + V_y$$

The per partnership probability of infection depends on the infectiousness and the likelihood that the partner is infected:

> **lambda := (B[1]*Y + B[2]*V[y])/NN;**

$$\lambda := \frac{B_1 Y + B_2 V_y}{X + V_x + Y + V_y}$$

We only care about the ultimate state, when the populations don't change: therefore we set all derivatives to zero.

> **equilibria := map(rhs,{eq1,eq2,eq3,eq4});**

$$\text{equilibria} := \left(\frac{c(B_1 Y + B_2 V_y) X}{X + V_x + Y + V_y} - (v + \mu) Y, \mu N_0 (1 - p) - \left(\frac{c(B_1 Y + B_2 V_y)}{X + V_x + Y + V_y} + \mu \right) X, \right. \\ \left. \mu N_0 p - \left(\frac{c(B_1 Y + B_2 V_y)}{X + V_x + Y + V_y} + \mu \right) V_x, \frac{c(B_1 Y + B_2 V_y) V_x}{X + V_x + Y + V_y} - (d + \mu) V_y \right)$$

Maple can solve those population equations, to give us solutions that look a little different than those in the paper but that we can experiment with:

> **sols := solve(equilibria, {X,Y,V[x],V[y]});**

$$\text{sols} := \left\{ \begin{array}{l} V_y = \frac{(-c B_2 p v + c B_1 d p - c B_2 \mu p + c B_1 \mu p + \mu^2 - c B_1 d + v \mu + d \mu + d v - c B_1 \mu) p N_0 \mu}{(d + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d)}, Y = - \\ \mu N_0 (-\mu^2 + \mu^2 p - c B_2 p^2 \mu + c B_1 p^2 \mu - 2 c B_1 \mu p + \mu p v + d \mu p + c B_2 \mu p - d \mu - v \mu + c B_1 \mu - c B_2 p^2 v \\ + c B_1 p^2 d + c B_2 p v + v d p - 2 c B_1 d p - d v + c B_1 d) / (-c B_1 v d + c B_1 v d p + v^2 d - c B_2 p v^2 + \mu^2 v \\ + \mu^2 d p - \mu^2 p v - \mu^2 c B_1 - \mu p v^2 + \mu d v + \mu^2 c B_1 p - \mu^2 c B_2 p + \mu c B_1 d p - 2 \mu c B_2 p v + \mu c B_1 p v \\ + \mu v d p - \mu c B_1 d - \mu c B_1 v + \mu v^2), \\ X = \frac{(\mu p - \mu + p^2 v - d + 2 d p - d p^2 - p v) \mu N_0}{-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d}, \\ V_x = - \frac{(\mu + p v - d p + d) p \mu N_0}{-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d} \end{array} \right\}, \\ \{Y = 0, X = N_0 - N_0 p, V_x = N_0 p, V_y = 0\}$$

One solution occurs if no-one is infected to start with, so $Y = V[y] = 0$. It is the other one we are interested in. It is curious that there is no solution if $X = V[x] = 0$; this is a result of an assumption in the paper that there is always a population of uninfected children "coming of age" to sexual maturity to supply uninfected susceptibles. (This is not pointed out in the paper).

Now let us check that the solutions we get are the same as those in the paper.

> **assign(sols[1]);**

> **Y;**

$$-\mu N_0 (-\mu^2 + \mu^2 p - c B_2 p^2 \mu + c B_1 p^2 \mu - 2 c B_1 \mu p + \mu p v + d \mu p + c B_2 \mu p - d \mu - v \mu + c B_1 \mu - c B_2 p^2 v)$$

$$\frac{+c B_1 p^2 d + c B_2 p v + v d p - 2 c B_1 d p - d v + c B_1 d}{(-c B_1 v d + c B_1 v d p + v^2 d - c B_2 p v^2 + \mu^2 v + \mu^2 d p - \mu^2 p v - \mu^2 c B_1 - \mu p v^2 + \mu d v + \mu^2 c B_1 p - \mu^2 c B_2 p + \mu c B_1 d p - 2 \mu c B_2 p v + \mu c B_1 p v + \mu v d p - \mu c B_1 d - \mu c B_1 v + \mu v^2)}$$

This doesn't look like the solution in the paper. We should see if we can simplify the computed Y to see if there is an error.

> **Y/c/N[0]/mu/(1-p)*(v+mu); # = lamstar/(c*lamstar+mu)**

$$\frac{-((- \mu^2 + \mu^2 p - c B_2 p^2 \mu + c B_1 p^2 \mu - 2 c B_1 \mu p + \mu p v + d \mu p + c B_2 \mu p - d \mu - v \mu + c B_1 \mu - c B_2 p^2 v + c B_1 p^2 d + c B_2 p v + v d p - 2 c B_1 d p - d v + c B_1 d)(v + \mu)}{((- c B_1 v d + c B_1 v d p + v^2 d - c B_2 p v^2 + \mu^2 v + \mu^2 d p - \mu^2 p v - \mu^2 c B_1 - \mu p v^2 + \mu d v + \mu^2 c B_1 p - \mu^2 c B_2 p + \mu c B_1 d p - 2 \mu c B_2 p v + \mu c B_1 p v + \mu v d p - \mu c B_1 d - \mu c B_1 v + \mu v^2) c (1 - p))}$$

We should get something like $\lambda/(c*\lambda+\mu)$, for a particular value of λ ---let us see if we can get something like the paper does.

> **solve(lstar/(c*lstar+mu)=%,lstar);;**

$$-\frac{-c B_2 p v + c B_1 d p - c B_2 \mu p + c B_1 \mu p + \mu^2 - c B_1 d + v \mu + d \mu + d v - c B_1 \mu}{c (\mu + p v - d p + d)}$$

> **lstar := %;**

$$lstar := -\frac{-c B_2 p v + c B_1 d p - c B_2 \mu p + c B_1 \mu p + \mu^2 - c B_1 d + v \mu + d \mu + d v - c B_1 \mu}{c (\mu + p v - d p + d)}$$

> **lstarf := map(factor,%);**

$$lstarf := -\frac{-c B_2 p v + c B_1 d p - c B_2 \mu p + c B_1 \mu p + \mu^2 - c B_1 d + v \mu + d \mu + d v - c B_1 \mu}{c (\mu + p v - d p + d)}$$

> **phieq := B[1]*(1-p)/(v+mu) + B[2]*p/(d+mu);**

$$phieq := \frac{B_1 (1 - p)}{v + \mu} + \frac{B_2 p}{d + \mu}$$

> **phi := 'phi';**

$$\phi := \phi$$

> **solve(phieq=phi,B[1]);**

$$-\frac{(-B_2 p + \phi d + \phi \mu)(v + \mu)}{(d + \mu)(p - 1)}$$

> **subs(B[1]=%,lstarf);**

$$\begin{aligned} & - \left(-c B_2 p v - \frac{c(-B_2 p + \phi d + \phi \mu)(v + \mu) d p}{(d + \mu)(p - 1)} - c B_2 \mu p - \frac{c(-B_2 p + \phi d + \phi \mu)(v + \mu) \mu p}{(d + \mu)(p - 1)} + \mu^2 \right. \\ & \left. + \frac{c(-B_2 p + \phi d + \phi \mu)(v + \mu) d}{(d + \mu)(p - 1)} + v \mu + d \mu + d v + \frac{c(-B_2 p + \phi d + \phi \mu)(v + \mu) \mu}{(d + \mu)(p - 1)} \right) / (c(\mu + p v - d p + d)) \end{aligned}$$

> **collect(%,phi);**

$$\begin{aligned} & - \frac{\left(-\frac{c(v + \mu) \mu p}{p - 1} - \frac{c(v + \mu) d p}{p - 1} + \frac{c(v + \mu) d}{p - 1} + \frac{c(v + \mu) \mu}{p - 1} \right) \phi}{c(\mu + p v - d p + d)} - \left(-c B_2 p v + \frac{c B_2 p^2 (v + \mu) d}{(d + \mu)(p - 1)} + \mu^2 - c B_2 \mu p \right. \\ & \left. + \frac{c B_2 p^2 (v + \mu) \mu}{(d + \mu)(p - 1)} + v \mu + d \mu - \frac{c B_2 p (v + \mu) d}{(d + \mu)(p - 1)} - \frac{c B_2 p (v + \mu) \mu}{(d + \mu)(p - 1)} + d v \right) / (c(\mu + p v - d p + d)) \end{aligned}$$

> **collect(%,phi,factor);**

$$\frac{(d + \mu)(v + \mu) \phi}{\mu + p v - d p + d} - \frac{(d + \mu)(v + \mu)}{(\mu + p v - d p + d) c}$$

This agrees with equations (6) and (8) of the paper (if we don't use $N=X+V[x]+Y+V[y]$) and even if we do it looks as though it is in the right form.

> **phi := phieq;**

$$\phi := \frac{B_1(1-p)}{v + \mu} + \frac{B_2 p}{d + \mu}$$

> **lamstar := %%;**

$$\text{lamstar} := \frac{(d + \mu)(v + \mu) \left(\frac{B_1(1-p)}{v + \mu} + \frac{B_2 p}{d + \mu} \right)}{\mu + p v - d p + d} - \frac{(d + \mu)(v + \mu)}{(\mu + p v - d p + d) c}$$

Now we compare our computed answer with the formula from the paper by subtracting the one from the other; if the difference is zero, then we know the paper result is correct.

> **Y - c*lamstar*mu*N[0]*(1-p)/(c*lamstar+mu)/(v+mu):**

> **normal(%);**

0

It is, so at least the formula for Y is right. But the formula for V[y] also looks different from that of the paper:

> **V[y];**

$$\frac{(-c B_2 p v + c B_1 d p - c B_2 \mu p + c B_1 \mu p + \mu^2 - c B_1 d + v \mu + d \mu + d v - c B_1 \mu) p N_0 \mu}{(d + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d)}$$

Again we subtract the formula from the paper and see if it is zero.

> **normal(V[y] - c*lamstar*mu*N[0]*p/(c*lamstar+mu)/(d+mu));**

0

It is. Now we see if we can recapitulate their argument to get the equilibrium population in the presence of infection:

> **normal(lamstar - c*lamstar*mu*N[0]*phi/(c*lamstar+mu)/Nstar):**

> **solve(%,Nstar):**

> **factor(%);**

$$\frac{(\mu + p v - d p + d) (-\mu B_2 p + \mu B_1 p - \mu B_1 - p B_2 v + B_1 d p - B_1 d) c \mu N_0}{(d + \mu) (v + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d)}$$

> **Nstar := %;**

$$Nstar := \frac{(\mu + p v - d p + d) (-\mu B_2 p + \mu B_1 p - \mu B_1 - p B_2 v + B_1 d p - B_1 d) c \mu N_0}{(d + \mu) (v + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d)}$$

Again, that formula looks different from the one in the paper, which defines things in terms of an auxiliary parameter, Omega.

> **Nstar = N[0]*(1-Omega)/(1-(1/c)*Omega/phi);**

$$\frac{(\mu + p v - d p + d) (-\mu B_2 p + \mu B_1 p - \mu B_1 - p B_2 v + B_1 d p - B_1 d) c \mu N_0}{(d + \mu) (v + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d)} =$$

$$1 - \frac{N_0 (1 - \Omega)}{c \left(\frac{B_1 (1-p)}{v + \mu} + \frac{B_2 p}{d + \mu} \right)}$$

> **tst := solve(%,Omega);**

$$tst := - \frac{\mu p v - v \mu - d \mu p - d v}{d v + d \mu + v \mu + \mu^2}$$

That again looks different from the formula in the paper:

> **tst - v*(1-p)/(v+mu) - d*p/(d+mu);**

$$- \frac{\mu p v - v \mu - d \mu p - d v}{d v + d \mu + v \mu + \mu^2} - \frac{v (1-p)}{v + \mu} - \frac{d p}{d + \mu}$$

> **normal(%);**

0

But it is not. Therefore, all the formulas in the paper are correct.

Now, we can extend the printed results and work out what the uninfected population NA would be:

> **eval(Nstar,p=0):**

> **NA := factor(%);**

$$NA := \frac{N_0 c B_1 \mu}{(c B_1 - v) (v + \mu)}$$

> **effectiveness_ratio := Nstar/NA;**

effectiveness_ratio :=

$$\frac{(\mu + p v - d p + d) (-\mu B_2 p + \mu B_1 p - \mu B_1 - p B_2 v + B_1 d p - B_1 d) (c B_1 - v)}{(d + \mu) (-\mu p v - c B_1 \mu + c B_1 \mu p - c B_2 \mu p + v \mu + d \mu p + d v + c B_1 d p - c B_2 p v - c B_1 d) B_1}$$

> **eval(%,p=1);**

$$\frac{(v + \mu) (-\mu B_2 - B_2 v) (c B_1 - v)}{(d + \mu) (-c B_2 \mu + d \mu + d v - c B_2 v) B_1}$$

> **normal(%);**

$$\frac{(c B_1 - v) B_2 (v + \mu)}{B_1 (c B_2 - d) (d + \mu)}$$

This short exploration showed an example of using Maple to verify the results in an area of mathematical biology and public health policy.